NAG Fortran Library Routine Document

F08JBF (DSTEVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JBF (DSTEVX) computes selected eigenvalues and, optionally, eigenvectors of a real symmetric tridiagonal matrix A. Eigenvalues and eigenvectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

2 Specification

```
      SUBROUTINE F08JBF (JOBZ, RANGE, N, D, E, VL, VU, IL, IU, ABSTOL, M, W,
     1                   Z, LDZ, WORK, IWORK, JFAIL, INFO)
      INTEGER            N, IL, IU, M, LDZ, IWORK(*), JFAIL(*), INFO
      double precision   D(*), E(*), VL, VU, ABSTOL, W(*), Z(LDZ,*), WORK(*)
      CHARACTER*1        JOBZ, RANGE
```

The routine may be called by its LAPACK name ***dstevx***.

3 Description

F08JBF (DSTEVX) computes the required eigenvalues and eigenvectors of A by reducing the tridiagonal matrix to diagonal form using the QR algorithm. Bisection is used to determine selected eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

2: RANGE – CHARACTER*1 *Input*

On entry: if RANGE = 'A', all eigenvalues will be found.

If RANGE = 'V', all eigenvalues in the half-open interval $(\mathrm{VL}, \mathrm{VU}]$ will be found.

If RANGE = 'I', the ILth to IUth eigenvalues will be found.

3: N – INTEGER *Input*

On entry: n, the order of the matrix.

Constraint: $\mathrm{N} \geq 0$.

4: D(*) – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: the n diagonal elements of the tridiagonal matrix A.

On exit: may be multiplied by a constant factor chosen to avoid over/underflow in computing the eigenvalues.

5: E(*) – ***double precision*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: the $(n-1)$ sub-diagonal elements of the tridiagonal matrix A.

On exit: may be multiplied by a constant factor chosen to avoid over/underflow in computing the eigenvalues.

6: VL – ***double precision*** *Input*
7: VU – ***double precision*** *Input*

On entry: if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.

Constraint: $\mathrm{VL} < \mathrm{VU}$.

If RANGE = 'A' or 'I', VL and VU are not referenced.

8: IL – INTEGER *Input*
9: IU – INTEGER *Input*

On entry: if RANGE = 'I', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If RANGE = 'A' or 'V', IL and IU are not referenced.

Constraints:

if $\mathrm{N} = 0$, $\mathrm{IL} = 1$ and $\mathrm{IU} = 0$;
if $\mathrm{N} > 0$, $1 \leq \mathrm{IL} \leq \mathrm{IU} \leq \mathrm{N}$.

10: ABSTOL – ***double precision*** *Input*

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval $[a, b]$ of width less than or equal to

$$\mathrm{ABSTOL} + \epsilon \max(|a|, |b|),$$

where ϵ is the ***machine precision***. If ABSTOL is less than or equal to zero, then $\epsilon\|A\|_1$ will be used in its place. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold $2 \times \mathrm{X02AMF}()$, not zero. If this routine returns with $\mathrm{INFO} > 0$, indicating that some eigenvectors did not converge, try setting ABSTOL to $2 \times \mathrm{X02AMF}()$. See Demmel and Kahan (1990).

11: M – INTEGER *Output*

On exit: the total number of eigenvalues found.

If RANGE = 'A', $\mathrm{M} = \mathrm{N}$.

If RANGE = 'I', $\mathrm{M} = \mathrm{IU} - \mathrm{IL} + 1$.

Constraint: $0 \leq \mathrm{M} \leq \mathrm{N}$.

12: W($*$) – ***double precision*** array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: the first M elements contain the selected eigenvalues in ascending order.

13: Z(LDZ,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{M})$.

On exit: if JOBZ = 'V', then if INFO = 0, the first m columns of Z contain the orthonormal eigenvectors of the matrix A corresponding to the selected eigenvalues, with the ith column of Z holding the eigenvector associated with $\mathrm{W}(i)$.

If an eigenvector fails to converge (INFO > 0), then that column of Z contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in JFAIL. If JOBZ = 'N', Z is not referenced.

Note: the user must ensure that at least $\max(1, \mathrm{M})$ columns are supplied in the array Z; if RANGE = 'V', the exact value of M is not known in advance and an upper bound must be used.

14: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08JBF (DSTEVX) is called.

Constraints:

if JOBZ = 'V', LDZ $\geq \max(1, \mathrm{N})$;
LDZ ≥ 1 otherwise.

15: WORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 5 \times \mathrm{N})$.

16: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, 5 \times \mathrm{N})$.

17: JFAIL($*$) – INTEGER array *Output*

Note: the dimension of the array JFAIL must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if INFO = 0, the first M elements of JFAIL are zero. If INFO > 0, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ = 'N', JFAIL is not referenced.

18: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

If INFO $= i$, then i eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A+E)$, where

$$\|E\|_2 = \mathrm{O}(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations is proportional to n^2 if JOBZ = 'N', and is proportional to n^3 if JOBZ = 'V' and RANGE = 'A', otherwise the number of floating point operations will depend upon the number of computed eigenvectors.

9 Example

To find the eigenvalues in the half-open interval $(0,5]$, and the corresponding eigenvectors, of the symmetric tridiagonal matrix

$$A = \begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 4 & 2 & 0 \\ 0 & 2 & 9 & 3 \\ 0 & 0 & 3 & 16 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08JBF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX
      PARAMETER        (NMAX=20,MMAX=10)
      INTEGER          LDZ
      PARAMETER        (LDZ=NMAX)
*     .. Local Scalars ..
      DOUBLE PRECISION ABSTOL, VL, VU
      INTEGER          I, IFAIL, IL, INFO, IU, J, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), E(NMAX), W(NMAX), WORK(5*NMAX),
     +                 Z(LDZ,MMAX)
      INTEGER          INDEX(NMAX), IWORK(5*NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AMF
      EXTERNAL         X02AMF
*     .. External Subroutines ..
      EXTERNAL         DSTEVX, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08JBF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the lower and upper bounds of the interval to be searched,
*        and read the diagonal and off-diagonal elements of the matrix
*        A from data file
*
         READ (NIN,*) VL, VU
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Set the absolute error tolerance for eigenvalues.  With ABSTOL
```

```
*        set to zero, the default value would be used instead
*
         ABSTOL = 2*X02AMF()
*
*        Solve the symmetric eigenvalue problem
*
         CALL DSTEVX('Vectors','Values in range',N,D,E,VL,VU,IL,IU,
     +               ABSTOL,M,W,Z,LDZ,WORK,IWORK,INDEX,INFO)
*
         IF (INFO.GE.0 .AND. M.LE.MMAX) THEN
*
*           Print solution
*
            WRITE (NOUT,99999) 'Number of eigenvalues found =', M
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99998) (W(J),J=1,M)
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,M,Z,LDZ,'Selected eigenvectors',
     +                  IFAIL)
            IF (INFO.GT.0) THEN
               WRITE (NOUT,99999)
     +           'INFO eigenvectors failed to converge, INFO =', INFO
               WRITE (NOUT,*)
     +           'Indices of eigenvectors that did not converge'
               WRITE (NOUT,99997) (INDEX(J),J=1,M)
            END IF
         ELSE IF (M.GT.MMAX) THEN
            WRITE (NOUT,99996) 'M greater than MMAX, M =', M,
     +        ', MMAX =', MMAX
         ELSE
            WRITE (NOUT,99999) 'Failure in DSTEVX. INFO =', INFO
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3X,(8F8.4))
99997 FORMAT (3X,(8I8))
99996 FORMAT (1X,A,I5,A,I5)
      END
```

9.2 Program Data

```
F08JBF Example Program Data

  4                    :Value of N

  0.0  5.0             :Values of VL and VU

  1.0  4.0  9.0  16.0  :End of diagonal elements
  1.0  2.0  3.0        :End of off-diagonal elements
```

9.3 Program Results

```
F08JBF Example Program Results

Number of eigenvalues found =    2

Eigenvalues
    0.6476  3.5470
Selected eigenvectors
         1       2
1   0.9396  0.3388
2  -0.3311  0.8628
3   0.0853 -0.3648
4  -0.0167  0.0879
```